



Superpowered game development.

Language Syntax

version 3.0.3916 beta Proposed Syntax

Live/current version at <http://SkookumScript.com/docs/>

Conan Reis (conan@AgogLabs.com)

February 10, 2017



Copyright © 2001-2017 Agog Labs Inc.
All Rights Reserved

Combined syntactical and lexical rules for SkookumScript in modified Extended Backus-Naur Form (EBNF). Production rules in *italics*. Terminals **coloured and in bold** and literal strings **quoted**. Optional groups: []. Repeating groups of zero or more: {}. Repeating groups of n or more: {}ⁿ. Mandatory groups: (). Alternatives (exclusive or): |. Disjunction (inclusive or): V.

Highlight colouring key: **in progress**, **planned**, **under consideration**.

File Names and Bodies:

```

method-filename1 = method-name '()' ['C'] '.sk'
method-file2    = ws {annotation wsr} parameters [ws code-block] ws

coroutine-filename = coroutine-name '()' ['C'] '.sk'
coroutine-file3  = ws {annotation wsr} parameter-list [ws code-block] ws

data-filename4   = '!Data' ['C'] '.sk'
data-file         = ws [data-definition {wsr data-definition} ws]
data-definition5 = {annotation wsr} [class-desc wsr] '!' data-name [ws binding]

annotation6      = '&' instance-name

object-id-filename7 = class-name ['-'] {printable} '.sk' '-' | '~' 'ids'
object-id-file8    = {ws symbol-literal | raw-object-id} ws
raw-object-id9     = {printable}1-255 end-of-line

enumeration-file = ws {enum-definition ws}
enum-definition  = enumeration-name ws [':' ws enum-class ws]
                  '[' ws [enumerator-defn {wsr enumerator-defn} ws] ']'
enumeration-name = '#' alphabetic {alphanumeric}
enumerator-defn10 = instance-name [ws ':' ws integer-literal]

flagset-file = ws {flagset-definition ws}
flagset-definition = flagset-name ws [':' ws flagset-class ws]
                   '[' ws [flag-definition {wsr flag-definition} ws] ']'

flag-definition11 = flag-name [ws ':' ws flag-operand]
flag-name          = instance-name
flag-operand12    = digits | flag-name | flag-op | flag-group
flag-group13      = '[' ws flag-op ws ']'
flag-op            = flag-operand ws flag-operator ws flag-operand
flag-operator      = logical-operator | '-'

```

¹ If optional '?' is used in query/predicate method name, use '-Q' as a substitute since question mark not valid in filename.

² Only immediate calls are permissible in the code block. If *code-block* is absent, it is defined in C++.

³ If *code-block* is absent, it is defined in C++.

⁴ A file name appended with 'C' indicates that the file describes class members rather than instance members. [Combine data files into one - add a keyword to separate instance and class and change name to "Class".]

⁵ *class-desc* is compiler hint for expected type of member variable. If class omitted, **Object** inferred or **Boolean** if *data-name* ends with '?'. If *data-name* ends with '?' and *class-desc* is specified it must be **Boolean** or *invoke-class with Boolean result type*. Optional binding part is default initialization and its result class can be used to infer member class. If default binding omitted, member must be bound to appropriate object before exiting instance or class constructor.

⁶ The context / file where an *annotation* is placed limits which values are valid.

⁷ Starts with the object id class name then optional source/origin tag (assuming a valid file title) - for example: Trigger-WorldEditor, Trigger-JoeDeveloper, Trigger-Extra, Trigger-Working, etc. A dash '-' in the file extension indicates an id file that is a compiler dependency and a tilde '~' in the file extension indicates that is not a compiler dependency

⁸ Note: if *symbol-literal* used for id then leading whitespace, escape characters and empty symbol ('') can be used.

⁹ Must have at least 1 character and may not have leading whitespace (ws), single quote (') nor *end-of-line* character.

¹⁰ Assigning an enumerator to an integer is discouraged though it is often handy to mirror underlying C++.

¹¹ If optional bit digit assignment used it is a 'persistent flag'. A flag assigned to another single flag is an 'aliased flag'. A flag assigned to a combination of flags using operations is a 'flag group'. If optional assignment is omitted, an unassigned bit is used.

¹² Valid digits range from 0 to 31 (i.e. 32-bits).

¹³ [flag-group could enclose any flag-operand, but grouping only has an effect around a flag-op, so this helps keep things tidy.]

Expressions:

expression = *literal* | *identifier* | *flow-control* | *primitive* | *invocation*

Literals:

literal = *boolean-literal* | *integer-literal* | *real-literal* | *string-literal* | *symbol-literal*
 | *char-literal* | *list-literal* | *closure* | *range-literal* | *closure-routine* | *map-literal*
 | *enumerator* | *flagset-literal*

boolean-literal = 'true' | 'false'

*integer-literal*¹ = ['−'] *digits-lead* ['r' *big-digit* {[*number-separator*] *big-digit*}]

*real-literal*² = ['−'] *digits-lead* V ('.' *digits-tail*) [*real-exponent*]

real-exponent = 'E' | 'e' ['−'] *digits-lead*

digits-lead = '0' | (non-zero-digit {[*number-separator*] *digit*})

digits-tail = *digit* {[*number-separator*] *digit*}

*number-separator*³ = '_'

string-literal = *escaped-string* | *raw-string* [ws '+' ws *string-literal*]

*escaped-string*⁴ = "" {character | ('\ ' [bracketed-args] *code-block*)} ""

*raw-string*⁵ = 'R' ['−' ['−']] "" {printable}¹⁰⁻¹⁶ 'C' {printable} ')' {printable}¹⁰⁻¹⁶ ""

symbol-literal = '' {character}¹⁰⁻²⁵⁵ ''

*char-literal*⁶ = '`' character

*list-literal*⁷ = [(*list-class* *constructor-name* *invocation-args*) | *class-desc*]
 '{' ws [*expression* {ws [','] ws *expression*} ws] '}'

*closure*⁸ = ('^' ['^'] ['_'] ws [*expression* ws]) V (*parameters* ws) *code-block*

*range-literal*⁹ = [*expression*] '..' [['.'] *expression*] | ('#' *expression*)

*closure-routine*¹⁰ = '^' *routine-identifier*

*map-literal*¹¹ = [(*map-class* *constructor-name* *invocation-args*) | (*class-desc* ':' ws [*class-desc* ws])]
 '{' ws (*key-value* {ws [','] ws *key-value*}) | ':' ws '}'

key-value = *expression* ws *binding*

*enumerator*¹² = (enum-class '.') | '#' *instance-name*

flagset-literal = (*flagset-class* '.') | '##' (*flag-name* | 'all' | 'none')

¹ 'r' indicates *digits-lead* is (r)adix/base from 1 to 36 - default 10 (decimal) if omitted. Ex: **2r** binary & **16r** hex. Valid *big-digit*(s) vary by the radix used. See *math-operator* footnote on how to differentiate subtract from negative *integer-literal*.

² Can use just *digits-lead* if **Real** type can be inferred from context otherwise the *digits-tail* fractional or *real-exponent* part is needed. See *math-operator* footnote on how to differentiate subtract from negative *real-literal*.

³ Visually separates parts of the number and ignored by the compiler. [Consider adding '_' since it will be used by C++.]

⁴ Escaped *code-block* indicates use of string interpolation with resulting object having **String()** conversion method called on it. If optional *bracket-args* present it is used as argument(s) to **String()** call.

⁵ Raw string using syntax similar to C++11. Optional '−' indicates initial & ending whitespace removed. Optional '−−' removes initial and ending whitespace and indentation of first line from all lines. Optional character sequence prior to 'C' used to make unique delimiter pair that must be matched with the closing character sequence following ')'.]

⁶ [Consider removing character literal and just using string literal esp. since UTF-8 character can be several combined glyphs. Also accent grave/backtick/backquote '`' is bad choice since it is **commonly used in Markdown, etc.** to demark sections of code.]

⁷ Item type determined via optional *list-class* constructor or specified class (or *class-desc* in the future). If neither supplied, then item type inferred using initial items, if no items then desired type used and if desired type not known then **Object** used.

⁸ [AKA code block/anonymous function/lambda expression] Optional '^', *parameters* or both must be provided (unless used in *closure-tail-args* where both optional). Optional *expression* (may not be *code-block*, *closure* or *routine-identifier*) captured and used as receiver/this for *code-block* - if omitted **this** inferred. Second optional '^' indicates scope of surrounding context used (i.e. refers to surrounding invoked object directly - which may go out of scope before this closure) rather than making a reference copy of any captured variables. Optional '_' indicates it is durational (like coroutine) - if not present durational/immediate inferred via *code-block*. Parameter types, return type, scope, whether surrounding **this** or temporary/parameter variables are used and captured may all be inferred if omitted.

⁹ [**first**]..**[.]last**[(#count)] Range from initial inclusive expression value (0/default? if omitted) to second exclusive expression value (-1/Type.max? if omitted, inclusive if optional third '.' used). If '#' used then until first expression + second expression. If neither expression is specified and the desired type is not known then **Integer** type is inferred.

¹⁰ Syntactical sugar/optimization of *closure* getting info such as interface from receiver object and single method/coroutine.

¹¹ Key-value types determined via optional *map-class* constructor or specified key-value *class-desc* types. If neither supplied, then key-value types inferred using initial *key-value* pairs, if no pairs then desired type used and if desired type not known then **Object** used for both key and value types.

¹² If desired enumeration class type can be inferred (like when passed as an argument) then optional enum-class may be omitted.

Identifiers:

<i>identifier</i> ¹	=	<i>variable-identifier</i> <i>reserved-identifier</i> <i>class-identifier</i> <i>object-id</i> <i>routine-identifier</i>
<i>variable-identifier</i> ²	=	<i>variable-name</i> ([<i>expression</i> ws <i>'.'</i> ws] <i>data-name</i>)
<i>variable-name</i>	=	<i>name-predicate</i>
<i>data-name</i> ³	=	<i>'@'</i> <i>'@@'</i> <i>variable-name</i>
<i>reserved-identifier</i>	=	<i>'nil'</i> <i>'this'</i> <i>'this_class'</i> <i>'this_code'</i>
<i>class-identifier</i>	=	<i>class-name</i> <i>enum-class</i> <i>flagset-class</i>
<i>object-id</i> ⁴	=	[<i>class-name</i>] <i>'@'</i> [<i>'?'</i> <i>'#'</i>] <i>symbol-literal</i>
<i>invoke-name</i>	=	<i>method-name</i> <i>coroutine-name</i>
<i>method-name</i> ⁵	=	<i>name-predicate</i> <i>constructor-name</i> <i>destructor-name</i> <i>class-name</i> <i>binary-operator</i> <i>postfix-operator</i>
<i>name-predicate</i> ⁶	=	<i>instance-name</i> [<i>'?'</i>]
<i>constructor-name</i>	=	<i>'!'</i> [<i>instance-name</i>]
<i>destructor-name</i> ⁷	=	<i>'!!'</i>
<i>coroutine-name</i>	=	<i>'_'</i> <i>instance-name</i>
<i>instance-name</i>	=	<i>lowercase</i> { <i>alphanumeric</i> }
<i>class-name</i>	=	<i>uppercase</i> { <i>alphanumeric</i> }
<i>routine-identifier</i>	=	<i>'@'</i> ([<i>expression</i>] <i>'.'</i>) <i>scope</i> <i>invoke-name</i>

Flow Control:

<i>flow-control</i>	=	<i>code-block</i> <i>conditional</i> <i>case</i> <i>when</i> <i>unless</i> <i>loop</i> <i>loop-exit</i> <i>loop-skip</i> <i>random</i> <i>concurrent</i> <i>class-cast</i> <i>class-conversion</i> <i>query-cast</i> <i>proviso</i> <i>return</i> <i>defer</i>
<i>code-block</i>	=	<i>'['</i> ws [<i>expression</i> { <i>wsr expression</i> } ws] <i>']'</i>
<i>conditional</i>	=	<i>'if'</i> {ws <i>expression</i> ws <i>code-block</i> } ¹⁺ [ws <i>else-block</i>]
<i>case</i>	=	<i>'case'</i> ws <i>expression</i> {ws <i>test-expr</i> ws <i>code-block</i> } ¹⁺ [ws <i>else-block</i>]
<i>else-block</i>	=	<i>'else'</i> ws <i>code-block</i>
<i>test-expr</i>	=	<i>case-operand</i> {ws [<i>','</i> ws] <i>case-operand</i> } ¹⁺
<i>case-operand</i>	=	<i>expression</i> <i>range-literal</i>
<i>when</i>	=	<i>expression</i> ws <i>'when'</i> ws <i>expression</i>
<i>unless</i>	=	<i>expression</i> ws <i>'unless'</i> ws <i>expression</i>
<i>loop</i> ⁸	=	<i>'loop'</i> [ws <i>instance-name</i>] ws <i>code-block</i>
<i>loop-exit</i> ⁹	=	<i>'exit'</i> [ws <i>instance-name</i>]
<i>loop-skip</i> ¹⁰	=	<i>'skip'</i> [ws <i>instance-name</i>]
<i>random</i> ¹¹	=	<i>'random'</i> [<i>'.'</i> <i>'unique'</i> <i>'mix'</i> <i>'remix'</i>] [<i>'('</i> ws <i>expression</i> ws <i>')'</i>] <i>any-tail</i> <i>weighted-tail</i>
<i>any-tail</i> ¹²	=	ws [<i>'['</i> ws { <i>expression</i> ws } ²⁺ <i>']'</i>]
<i>weighted-tail</i> ¹³	=	{ws <i>expression</i> ws <i>code-block</i> } ²⁺

¹ Scoping not necessary - instance names may not be overridden and classes and implicit identifiers effectively have global scope.

² Optional *expression* can be used to access data member from an object - if omitted, *this* is inferred.

³ *'@'* indicates instance data member and *'@@'* indicates class instance data member.

⁴ If *class-name* absent, **Actor** inferred or desired type if known. If optional *'?'* present and object not found at runtime then result is *nil* else assertion error occurs. Optional *'#'* indicates no lookup - just return name identifier validated by class type.

⁵ A method using *class-name* allows explicit conversion similar to *class-conversion* except that the method is always called. [Consider: could also be used as a mechanism for custom literals - ex: *"identifier".CustomType* or *42.GameId*.]

⁶ Optional *'?'* used as convention to indicate predicate variable or method of return type **Boolean** (*true* or *false*).

⁷ Destructor calls are only valid in the scope of another destructor's code block. [Ensure compiler check.]

⁸ The optional *instance-name* names the loop for specific reference by a *loop-exit* which is useful for nested loops.

⁹ A *loop-exit* is valid only in the code block scope of the loop that it references.

¹⁰ Restarts/continues loop by jumping to loop start - valid only in the code block scope of the loop that it references.

¹¹ Only chosen path is evaluated. Optional modifier after *'.'* has meanings: *'unique'* - the previous flow path is not repeated; *'mix'* - the paths are randomized once initially and iterated through in sequence repeating; *'remix'* - similar to *'mix'* but paths are randomized after each full pass and the first new path is guaranteed not to be the same as the last path in the previous sequence. Optional *expression* in brackets *'('* is **Random** object to use and if absent the default random generator is used.

¹² Any *expression* is evaluated at random with a uniform distribution taking any modifier into consideration.

¹³ The *expression* represents a **Real** or **Integer** value for the weighted probability (value / sum of values) for that flow path. The sum of values need not add up to 1, 100, or any other specific value. A value of *<=0* omits that path in that particular evaluation.

```

concurrent      = sync | race | rush | branch | divert
sync1          = 'sync' ws code-block
race2          = 'race' ws code-block
rush3          = 'rush' ws code-block
branch4        = 'branch' ws expression
divert5        = 'divert' ws [expression ws] code-block
return6        = 'return' ws expression
defer7         = 'defer' ws expression
query-cast8    = expression ws '<?>' {ws class-desc [ws code-block]}1+ [ws else-block]
proviso9       = '\\proviso' wsr proviso-test ws code-block
proviso-test10 = instance-name | ('[' proviso-test ']') | operator-call

```

Primitives:

```

primitive       = create-temporary | bind | class-cast | class-conversion | nil-coalesce | list-expansion
create-temporary = define-temporary [ws binding]
define-temporary = '!' ws variable-name
bind11          = variable-identifier ws binding
binding12       = ':' ws expression
class-cast13    = expression ws '<>' [class-desc]
class-conversion14 = expression ws '>>' [class-name]
nil-coalesce15 = expression ws '??' ws expression
list-expansion  = '%' expression

```

¹ 2+ durational expressions run concurrently and next *expression* executed when *all* expressions returned (result **nil**, return args bound in order of expression completion).

² 2+ durational expressions run concurrently and next *expression* executed when *fastest* expression returns (result **nil**, return args of fastest expression bound) and other expressions are *aborted*.

³ Like *race* except: return args bound in expression completion order and other expressions continue until *completed*. *code-block* is essentially a closure with captured temporary variables to ensure temporal scope safety.

⁴ Durational expression run concurrently with surrounding context and the next *expression* executed immediately (result **InvokedCoroutine**). *expression* is essentially a closure with captured temporary variables to ensure temporal scope safety. Any return arguments will be bound to the captured variables.

⁵ Rather than inheriting that caller's updater **Mind** object, durational expressions in block are updated by **the mind object specified by the optional expression**. If the optional expression is not specified then the **Master Mind** object is inferred.

⁶ Like *race* except: return args bound in expression completion order and other expressions continue until *completed*. *code-block* is essentially a closure with captured temporary variables to ensure temporal scope safety.

⁷ Like *race* except: return args bound in expression completion order and other expressions continue until *completed*. *code-block* is essentially a closure with captured temporary variables to ensure temporal scope safety.

⁸ if *expression* is a *variable-identifier* its type is modified in any matching clause block. If a clause block is omitted the result of *expression* is cast to the matching type and given as a result.

⁹ Conditional code that will be compiled only if *proviso-test* evaluates to true. [Alternatively, this could be structured like a *conditional expression* with 1+ test clauses and an optional "else" clause.]

¹⁰ *instance-name* refers to set of predefined proviso labels - example "debug", "extra_check", etc. [It could be any valid Boolean *expression* - with limits based on availability of code at compile time.] *operator-call* uses *proviso-test* rather than *expression*.

¹¹ [Consider: Make *bind* valid only in a *code-block* so that it is not confused in *key-value* for *map-literal*.] Compiler gives warning if *bind* used in *code-block* of a *closure* since it will be binding to captured variable not original variable in surrounding context.

¹² [Stylistically prefer no ws prior to ':' - though not enforcing it via compiler.]

¹³ Compiler *hint* that expression evaluates to specified class - otherwise error. *class-desc* optional if desired type can be inferred. If *expression* is *variable-identifier* then parser updates type context. [Debug: runtime ensures class specified is received. Release: no code generated.]

¹⁴ Explicit conversion to specified class. *class-name* optional if desired type inferable. Ex: **42>>String** calls convert method **Integer@String()** i.e. **42.String()** - whereas **"hello">>String** generates no extra code and is equivalent to **"hello"**.

¹⁵ **expr1??expr2** is essentially equivalent to **if expr1.nil? [expr2] else [expr1<>TypeNoneRemoved]**.

Invocations:

<i>invocation</i>	=	<i>invoke-call</i> <i>invoke-cascade</i> <i>apply-operator</i> <i>invoke-operator</i> <i>index-operator</i> <i>slice-operator</i> <i>instantiation</i>
<i>invoke-call</i> ¹	=	(<i>[expression ws '.' ws]</i> <i>invoke-selector</i>) <i>operator-call</i>
<i>invoke-cascade</i>	=	<i>expression ws '.' ws '[' {ws <i>invoke-selector</i> <i>operator-selector</i>}²⁺ ws ']'</i>
<i>apply-operator</i> ²	=	<i>expression ws '%' '%>' '%,' '%<' '%.'</i> <i>invoke-selector</i>
<i>invoke-operator</i> ³	=	<i>expression bracketed-args</i>
<i>index-operator</i> ⁴	=	<i>expression '{' ws expression ws '}' [ws <i>binding</i>]</i>
<i>slice-operator</i> ⁵	=	<i>expression '{' ws range-literal [wsr <i>expression</i>] ws '}'</i>
<i>instantiation</i> ⁶	=	<i>[class-instance] expression '!' [instance-name] invocation-args</i>
<i>invoke-selector</i>	=	<i>[scope] invoke-name invocation-args</i>
<i>scope</i>	=	<i>class-unary '@'</i>
<i>operator-call</i> ⁷	=	(<i>prefix-operator ws expression</i>) (<i>expression ws operator-selector</i>)
<i>operator-selector</i>	=	<i>postfix-operator</i> (<i>binary-operator ws expression</i>)
<i>prefix-operator</i> ⁸	=	<i>'not'</i> <i>'-'</i>
<i>binary-operator</i>	=	<i>math-operator</i> <i>compare-op</i> <i>logical-operator</i> <i>':='</i>
<i>math-operator</i> ⁹	=	<i>'+' '+=' '-' '-=' '*' '*=' '/' '/='</i>
<i>compare-op</i>	=	<i>'=' '~=' '>' '>=' '<' '<='</i>
<i>logical-operator</i> ¹⁰	=	<i>'and' 'or' 'xor' 'nand' 'nor' 'nxor'</i>
<i>postfix-operator</i>	=	<i>'++' '--'</i>
<i>invocation-args</i> ¹¹	=	<i>[bracketed-args] closure-tail-args</i>
<i>bracketed-args</i>	=	<i>(' ws [send-args ws] [';' ws return-args ws] ')'</i>
<i>closure-tail-args</i> ¹²	=	<i>ws send-args ws closure [ws ';' ws return-args]</i>
<i>send-args</i>	=	<i>[argument] {ws [';' ws] [argument]}</i>
<i>return-args</i>	=	<i>[return-arg] {ws [';' ws] [return-arg]}</i>
<i>argument</i>	=	<i>[named-spec ws] expression</i>
<i>return-arg</i> ¹³	=	<i>[named-spec ws] variable-identifier define-temporary</i>
<i>named-spec</i> ¹⁴	=	<i>variable-name '#'</i>

¹ If an *invoke-call*'s optional *expression* (the receiver) is omitted, *'this.'* is implicitly inferred. *[Consider whitespace.]*

² If **List**, each item (or none if empty) sent call - coroutines called using *%- sync*, *%> - race*, *%,- - rush*, *%<- split*, *%,- span* respectively and returns itself (the list). If non-list it executes like a normal invoke call - i.e. *'%'* is synonymous to *'.'* except that if *nil* the call is ignored, then the normal result or *nil* respectively is returned.

³ Akin to *expr.invoke(...)* or *expr._invoke(...)* depending if *expression* immediate or durational - **and** if enough context is available the arguments are compile-time type-checked plus adding any default arguments.

⁴ Gets item (or sets item if *binding* present) at specified index object. Syntactic sugar for *at()* or *at_set()*.

⁵ Returns **Integer** sub-range: *{[first]..[.]last}[#count][step]}*. Where: **last** and **first** may be negative with *-1* last item, *-2* penultimate item, etc.; **step** may be negative indicating sub-range in reverse order.

⁶ If *class-instance* can be inferred then it may be omitted, *expression* used rather than *class-instance* provides lots of syntactic sugar: *expr!ctor()* is alias for *ExprClass!ctor(expr)* - ex: *num!copy* equals *Integer!copy(num)*; brackets are optional for *invocation-args* if it can have just the first argument; a constructor-name of *!* is an alias for *!copy* - ex: *num!* equals *Integer!copy(num)*; and if *expr!ident* does not match a constructor it will try *ExprClass!copy(expr).ident* - ex: *str!uppercase* equals *String!copy(str).uppercase*.

⁷ Every operator has a named equivalent. For example *:=* and *assign()*. Operators do **not** have special order of precedence - any order other than left to right must be indicated by using code block brackets (*[* and *]*).

⁸ See math-operator footnote about subtract on how to differentiate from a negation *'-'* prefix operator.

⁹ In order to be recognized as single subtract *'-'* expression and not an *expression* followed by a second *expression* starting with a minus sign, the minus symbol *'-'* must either have whitespace following it or no whitespace on either side.

¹⁰ Like other identifiers - whitespace is required when next to other identifier characters.

¹¹ *bracketed-args* may be omitted if the invocation can have zero arguments

¹² Routines with last send parameter as mandatory closure may omit brackets *'()'* and closure arguments may be simple *code-block* (omitting *'\n'* and parameters and inferring from parameter). Default arguments indicated via comma *','* separators.

¹³ If a temporary is defined in the *return-arg*, it has scope for the entire surrounding code block.

¹⁴ Used at end of argument list and only followed by other named arguments. Use compatible **List** object for group argument. Named arguments evaluated in parameter index order regardless of call order since defaults may reference earlier parameters.

Parameters:

```

parameters1      = parameter-list [ws class-desc] ['!']
parameter-list    = '(' ws [send-params ws] [';' ws return-params ws] ')'
send-params       = parameter {ws [';' ws] parameter}
return-params     = return-param {ws [';' ws] return-param}
parameter         = unary-param | group-param
return-param      = param-specifier | group-specifier
unary-param2     = param-specifier [ws binding]
param-specifier3 = [class-desc wsr] variable-name ['!']
group-param4    = group-specifier [ws binding]
group-specifier5 = '{' ws [class-desc {wsr class-desc} ws] '}' [digits] ws instance-name

```

Class Descriptors:

```

class-desc        = class-unary | class-union | nested-enum | label
class-unary       = class-instance | meta-class | enum-class | flagset-class
class-instance    = class | list-class | invoke-class | map-class | code-class
meta-class        = '<' class-name '>'
class-union6     = '<' class-unary {'|' class-unary}1+ '>'
invoke-class7   = ['_' | '+'] parameters
list-class8      = List '{' ws [class-desc ws] '}'
map-class9       = Map '{' ws [class-desc] ':' ws [class-desc ws] '}'
code-class10    = [class-unary ws] '.' invoke-class
enum-class11    = [class-name ['@' invoke-name]] enumeration-name
nested-enum       = '#' | enumeration-name ws enumerator-list
label            = '#' 'Symbol' | 'String'
flagset-class     = [class-name] flagset-name
flagset-name      = '##' alphabetic {alphanumeric}

```

¹ Optional *class-desc* is return class - if type not specified **Object** is inferred (or **Boolean** type for predicates or **Auto_** type for closures) for nested parameters / code blocks and **InvokedCoroutine** is inferred for coroutine parameters. **!** indicates result returned by value (**!copy()** is called on it) rather than just being returned by reference.

² The optional *binding* indicates the parameter has a default argument (i.e. supplied *expression*) when argument is omitted. **~** uses instance scope and **::** indicates calling scope used to evaluate the default.

³ **!** indicates arguments passed by value (**!copy()** is called on them) rather than just being passed by reference. If optional *class-desc* is omitted **Object** is inferred (or **Auto_** for closures or **Boolean** if *variable-name* ends with **?**). If *variable-name* ends with **?** and *class-desc* is specified it must be **Boolean**.

⁴ If default *binding* is omitted an empty list is used as the default.

⁵ **Object** inferred if no classes specified. Class of resulting list bound to *instance-name* is class union of all classes specified. The optional *digits* indicates the minimum number of arguments that must be present.

⁶ Indicates that the class is any one of the classes specified and which in particular is not known at compile time.

⁷ **_** indicates durational (like coroutine), **+** indicates durational/immediate and lack of either indicates immediate (like method). Class **Closure** matches any closure interface. Identifiers and defaults used for parameterless closure arguments.

⁸ **List** is any **List** derived class. If *class-desc* in item class descriptor is omitted, **Object** is inferred when used as a type or the item type is deduced when used with a *list-literal*. A *list-class* of any item type can be passed to a simple untyped **List** class.

⁹ **Map** is any **Map** derived class. If *class-desc* in key/value class descriptors is omitted, **Object** inferred when used as type or types are deduced when used with *map-literal*. A *map-class* of any key/value type can be passed to simple untyped **Map** class.

¹⁰ Optional *class-unary* is the receiver type of the method/coroutine - if it is omitted then **Object** is inferred.

¹¹ Optional *class-name* and *invoke-name* qualification only needed if it cannot be inferred from the context - so it may be omitted and inferred if inside the required scope or if the expected enumeration class type is known, etc.

Whitespace:

`wsr`¹ = `{ whitespace }1+`
`ws` = `{ whitespace }`
`whitespace` = `whitespace-char | comment`
`whitespace-char` = `' ' | formfeed | newline | carriage-return | horiz-tab | vert-tab`
`end-of-line` = `newline | carriage-return | end-of-file`
`comment` = `single-comment | multi-comment | parser-comment`
`single-comment` = `'/' { printable } end-of-line`
`multi-comment` = `/* { printable } [multi-comment { printable }] */`
`parser-comment`² = `'\ ' *parser-hint* end-of-line`

Characters and Digits:

`character` = `escape-sequence | printable`
`escape-sequence`³ = `'\ ' integer-literal | printable`
`alphanumeric` = `alphabetic | digit | '_'`
`alphabetic` = `uppercase | lowercase`
`lowercase` = `'a' | ... | 'z'`
`uppercase` = `'A' | ... | 'Z'`
`digits` = `'0' | (non-zero-digit {digit})`
`digit` = `'0' | non-zero-digit`
`non-zero-digit` = `'1' | '2' | '3' | '4' | '5' | '6' | '7' | '8' | '9'`
`big-digit` = `digit | alphabetic`

¹ `wsr` is an abbreviation for (w)hite (s)pace (r)equied.

² [Consider different compiler hints - ex: disable warning X. Should also be a way to hook in application custom compiler hints.]

³ Special escape characters: `'n'` - newline, `'t'` - tab, `'v'` - vertical tab, `'b'` - backspace, `'r'` - carriage return, `'f'` - formfeed, and `'a'` - alert. All other characters resolve to the same character including `'\'`, `'"`, and `'"`. Also see *escaped-string*.