



Superpowered game development.

Language Syntax

version 3.0.3916 beta

Live/current version at <http://SkookumScript.com/docs/>

Conan Reis (conan@AgogLabs.com)

February 10, 2017



Copyright © 2001-2017 Agog Labs Inc.
All Rights Reserved

Combined syntactical and lexical rules for SkookumScript in modified Extended Backus-Naur Form (EBNF). Production rules in *italics*. Terminals **coloured and in bold** and literal strings **quoted**. Optional groups: []. Repeating groups of zero or more: {}. Repeating groups of n or more: {}ⁿ. Mandatory groups: (). Alternatives (exclusive or): |. Disjunction (inclusive or): V.

File Names and Bodies:

*method-filename*¹ = *method-name* '()' ['C'] '.sk'
*method-file*² = ws {*annotation* *wsr*} *parameters* [ws *code-block*] ws
coroutine-filename = *coroutine-name* '()' '.sk'
*coroutine-file*³ = ws {*annotation* *wsr*} *parameter-list* [ws *code-block*] ws
*data-filename*⁴ = '!Data' ['C'] '.sk'
data-file = ws [*data-definition* {*wsr* *data-definition*} ws]
*data-definition*⁵ = {*annotation* *wsr*} [*class-desc* *wsr*] '!' *data-name*
*annotation*⁶ = '&' *instance-name*
*object-id-filename*⁷ = *class-name* ['-'] {**printable**} '.sk' '-' | '~' 'ids'
*object-id-file*⁸ = {ws *symbol-literal* | *raw-object-id*} ws
*raw-object-id*⁹ = {**printable**}¹⁻²⁵⁵ *end-of-line*

Expressions:

expression = *literal* | *identifier* | *flow-control* | *primitive* | *invocation*

Literals:

literal = *boolean-literal* | *integer-literal* | *real-literal* | *string-literal* | *symbol-literal*
 | *char-literal* | *list-literal* | *closure*
boolean-literal = 'true' | 'false'
*integer-literal*¹⁰ = ['-'] *digits-lead* ['r' *big-digit* {[*number-separator*] *big-digit*}]
*real-literal*¹¹ = ['-'] *digits-lead* V ('.' *digits-tail*) [*real-exponent*]
real-exponent = 'E' | 'e' ['-'] *digits-lead*
*digits-lead*¹² = '0' | (non-zero-digit {['-'] *digit*})
digits-tail = *digit* {['-'] *digit*})
string-literal = *simple-string* {ws '+' ws *simple-string*}
simple-string = '"' {*character*} '"'
symbol-literal = "'" {*character*}⁰⁻²⁵⁵ "'"
char-literal = '`' *character*
*list-literal*¹³ = [(*list-class* *constructor-name* *invocation-args*) | *class-name*]
 {' ' ws [*expression* {ws [' , ' ws] *expression*} ws] '}'

¹ If optional '?' is used in query/predicate method name, use '-Q' as a substitute since question mark not valid in filename.

² Only immediate calls are permissible in the code block. If *code-block* is absent, it is defined in C++.

³ If *code-block* is absent, it is defined in C++.

⁴ A file name appended with 'C' indicates that the file describes class members rather than instance members.

⁵ *class-desc* is compiler hint for expected type of member variable. If class omitted, **Object** inferred or **Boolean** if *data-name* ends with '?'. If *data-name* ends with '?' and *class-desc* is specified it must be **Boolean**.

⁶ The context / file where an *annotation* is placed limits which values are valid.

⁷ Starts with the object id class name then optional source/origin tag (assuming a valid file title) - for example: Trigger-WorldEditor, Trigger-JoeDeveloper, Trigger-Extra, Trigger-Working, etc. A dash '-' in the file extension indicates an id file that is a compiler dependency and a tilde '~' in the file extension indicates that is not a compiler dependency

⁸ Note: if *symbol-literal* used for id then leading whitespace, escape characters and empty symbol ('') can be used.

⁹ Must have at least 1 character and may not have leading whitespace (ws), single quote ('') nor *end-of-line* character.

¹⁰ 'r' indicates *digits-lead* is (r)adix/base from 1 to 36 - default 10 (decimal) if omitted. Ex: **2r** binary & **16r** hex. Valid *big-digit*(s) vary by the radix used. See *math-operator* footnote on how to differentiate subtract from negative *integer-literal*.

¹¹ Can use just *digits-lead* if **Real** type can be inferred from context otherwise the *digits-tail* fractional or *real-exponent* part is needed. See *math-operator* footnote on how to differentiate subtract from negative *real-literal*.

¹² '-' visually separates parts of the number and ignored by the compiler.

¹³ Item type determined via optional *list-class* constructor or specified class. If neither supplied, then item type inferred using initial items, if no items then **object** used.

*closure*¹ = (**'^'** [**'_'** ws] [expression ws]) V (parameters ws) code-block

Identifiers:

*identifier*² = variable-identifier | reserved-identifier | class-name | object-id
*variable-identifier*³ = variable-name | ([expression ws **'.'** ws] data-name)
variable-name = name-predicate
*data-name*⁴ = **'@'** | **'@@'** variable-name
reserved-identifier = **'nil'** | **'this'** | **'this_class'** | **'this_code'**
*object-id*⁵ = [class-name] **'@'** [**'?'** | **'#'**] symbol-literal
invoke-name = method-name | coroutine-name
*method-name*⁶ = name-predicate | constructor-name | destructor-name | class-name
*name-predicate*⁷ = instance-name [**'?'**]
constructor-name = **'!'** [instance-name]
*destructor-name*⁸ = **'!!'**
coroutine-name = **'_'** instance-name
instance-name = lowercase {alphanumeric}
class-name = uppercase {alphanumeric}

Flow Control:

flow-control = code-block | conditional | case | when | unless | | loop | loop-exit | concurrent
 | class-cast | class-conversion
code-block = [**'['** ws [expression {wsr expression} ws] **']'**
conditional = **'if'** {ws expression ws code-block}¹⁺ [ws else-block]
case = **'case'** ws expression {ws expression ws code-block}¹⁺ [ws else-block]
else-block = **'else'** ws code-block
when = expression ws **'when'** ws expression
unless = expression ws **'unless'** ws expression
*loop*⁹ = **'loop'** [ws instance-name] ws code-block
*loop-exit*¹⁰ = **'exit'** [ws instance-name]
concurrent = sync | race | branch | divert
*sync*¹¹ = **'sync'** ws code-block
*race*¹² = **'race'** ws code-block
*branch*¹³ = **'branch'** ws expression
*divert*¹⁴ = **'divert'** ws code-block

¹ [AKA code block/anonymous function/lambda expression] Optional **'^'**, parameters or both must be provided (unless used in *closure-tail-args* where both optional). Optional *expression* (may not be *code-block*, *closure* or *routine-identifier*) captured and used as receiver/this for *code-block* - if omitted **this** inferred. Optional **'_'** indicates it is durational (like coroutine) - if not present durational/immediate inferred via *code-block*. Parameter types, return type, scope, whether surrounding **this** or temporary/parameter variables are used and captured may all be inferred if omitted.

² Scoping not necessary - instance names may not be overridden and classes and implicit identifiers effectively have global scope.

³ Optional *expression* can be used to access data member from an object - if omitted, **this** is inferred.

⁴ **'@'** indicates instance data member and **'@@'** indicates class instance data member.

⁵ If *class-name* absent, **Actor** inferred or desired type if known. If optional **'?'** present and object not found at runtime then result is **nil** else assertion error occurs. Optional **'#'** indicates no lookup - just return name identifier validated by class type.

⁶ A method using *class-name* allows explicit conversion similar to *class-conversion* except that the method is always called.

⁷ Optional **'?'** used as convention to indicate predicate variable or method of return type **Boolean (true or false)**.

⁸ Destructor calls are only valid in the scope of another destructor's code block.

⁹ The optional *instance-name* names the loop for specific reference by a *loop-exit* which is useful for nested loops.

¹⁰ A *loop-exit* is valid only in the code block scope of the loop that it references.

¹¹ 2+ durational expressions run concurrently and next *expression* executed when *all* expressions returned (result **nil**, return args bound in order of expression completion).

¹² 2+ durational expressions run concurrently and next *expression* executed when *fastest* expression returns (result **nil**, return args of fastest expression bound) and other expressions are *aborted*.

¹³ Durational expression run concurrently with surrounding context and the next *expression* executed immediately (result **InvokedCoroutine**). *expression* is essentially a closure with captured temporary variables to ensure temporal scope safety. Any return arguments will be bound to the captured variables.

¹⁴ Durational expressions in block are updated by the **Master Mind** rather than inheriting that caller's updater **Mind** object.

Invocations:

<i>invocation</i>	=	<i>invoke-call</i> <i>invoke-cascade</i> <i>apply-operator</i> <i>invoke-operator</i> <i>index-operator</i> <i>instantiation</i>
<i>invoke-call</i> ¹	=	(<i>[expression ws '.' ws]</i> <i>invoke-selector</i>) <i>operator-call</i>
<i>invoke-cascade</i>	=	<i>expression ws '.' ws '[' {ws <i>invoke-selector</i> <i>operator-selector</i>}²⁺ ws ']'</i>
<i>apply-operator</i> ²	=	<i>expression ws '%' '%>' invoke-selector</i>
<i>invoke-operator</i> ³	=	<i>expression bracketed-args</i>
<i>index-operator</i> ⁴	=	<i>expression '{' ws expression ws '}' [ws binding]</i>
<i>instantiation</i> ⁵	=	<i>class-instance</i> <i>expression '!' [instance-name] invocation-args</i>
<i>invoke-selector</i>	=	<i>[scope] invoke-name invocation-args</i>
<i>scope</i>	=	<i>class-name '@'</i>
<i>operator-call</i> ⁶	=	(<i>prefix-operator ws expression</i>) (<i>expression ws operator-selector</i>)
<i>operator-selector</i>	=	<i>postfix-operator</i> (<i>binary-operator ws expression</i>)
<i>prefix-operator</i> ⁷	=	'not' '-'
<i>binary-operator</i>	=	<i>math-operator</i> <i>compare-op</i> <i>logical-operator</i> ':='
<i>math-operator</i> ⁸	=	'+' '+=' '-' '-=' '*' '*=' '/' '/='
<i>compare-op</i>	=	'=' '~=' '>' '>=' '<' '<='
<i>logical-operator</i> ⁹	=	'and' 'or' 'xor' 'nand' 'nor' 'nxor'
<i>postfix-operator</i>	=	'++' '--'
<i>invocation-args</i> ¹⁰	=	<i>[bracketed-args]</i> <i>closure-tail-args</i>
<i>bracketed-args</i>	=	'(' ws <i>[send-args ws]</i> '['; ws <i>return-args ws]</i> ')'
<i>closure-tail-args</i> ¹¹	=	<i>ws send-args ws closure [ws ';' ws return-args]</i>
<i>send-args</i>	=	<i>[argument] {ws '[' ws [argument]}</i>
<i>return-args</i>	=	<i>[return-arg] {ws '[' ws [return-arg]}</i>
<i>argument</i>	=	<i>[named-spec ws] expression</i>
<i>return-arg</i> ¹²	=	<i>[named-spec ws] variable-identifier define-temporary</i>
<i>named-spec</i> ¹³	=	<i>variable-name '#'</i>

¹ If an *invoke-call*'s optional *expression* (the receiver) is omitted, '**this.**' is implicitly inferred.

² If **List**, each item (or none if empty) sent call - coroutines called using **%-sync**, **%>-race** respectively and returns itself (the list). If non-list it executes like a normal invoke call - i.e. **%** is synonymous to **.** except that if **nil** the call is ignored, then the normal result or **nil** respectively is returned.

³ Akin to **expr.invoke(...)** or **expr._invoke(...)** depending if *expression* immediate or durational - *and* if enough context is available the arguments are compile-time type-checked plus adding any default arguments.

⁴ Gets item (or sets item if *binding* present) at specified index object. Syntactic sugar for **at()** or **at_set()**.

⁵ *expression* used rather than *class-instance* provides lots of syntactic sugar: **expr!ctor()** is alias for **ExprClass!ctor(expr)** - ex: **num!copy** equals **Integer!copy(num)**; brackets are optional for *invocation-args* if it can have just the first argument; a constructor-name of **!** is an alias for **!copy** - ex: **num!** equals **Integer!copy(num)**; and if **expr!ident** does not match a constructor it will try **ExprClass!copy(expr).ident** - ex: **str!uppercase** equals **String!copy(str).uppercase**.

⁶ Every operator has a named equivalent. For example **:=** and **assign()**. Operators do *not* have special order of precedence - any order other than left to right must be indicated by using code block brackets (**[** and **]**).

⁷ See math-operator footnote about subtract on how to differentiate from a negation **'-'** prefix operator.

⁸ In order to be recognized as single subtract **'-'** expression and not an *expression* followed by a second *expression* starting with a minus sign, the minus symbol **'-'** must either have whitespace following it or no whitespace on either side.

⁹ Like other identifiers - whitespace is required when next to other identifier characters.

¹⁰ *bracketed-args* may be omitted if the invocation can have zero arguments

¹¹ Routines with last send parameter as mandatory closure may omit brackets **'()'** and closure arguments may be simple *code-block* (omitting **'\n'** and parameters and inferring from parameter). Default arguments indicated via comma **' , '** separators.

¹² If a temporary is defined in the *return-arg*, it has scope for the entire surrounding code block.

¹³ Used at end of argument list and only followed by other named arguments. Use compatible **List** object for group argument. Named arguments evaluated in parameter index order regardless of call order since defaults may reference earlier parameters.

Primitives:

primitive = *create-temporary* | *bind* | *class-cast* | *class-conversion*
create-temporary = *define-temporary* [ws *binding*]
define-temporary = **'!**' ws *variable-name*
*bind*¹ = *variable-identifier* ws *binding*
*binding*² = **'::'** ws *expression*
*class-cast*³ = *expression* ws **'<>'** [*class-desc*]
*class-conversion*⁴ = *expression* ws **'>>'** [*class-name*]

Parameters:

*parameters*⁵ = *parameter-list* [ws *class-desc*]
parameter-list = **'('** ws [*send-params* ws] [**';**' ws *return-params* ws] **)'**
send-params = *parameter* {ws [**','**' ws] *parameter*}
return-params = *param-specifier* {ws [**','**' ws] *param-specifier*}
parameter = *unary-param* | *group-param*
*unary-param*⁶ = *param-specifier* [ws *binding*]
*param-specifier*⁷ = [*class-desc* wsr] *variable-name*
group-param = *group-specifier*
*group-specifier*⁸ = **'{'** ws [*class-desc* {wsr *class-desc*} ws] **'}'** ws *instance-name*

Class Descriptors:

class-desc = *class-unary* | *class-union*
class-unary = *class-instance* | *meta-class*
class-instance = *class-name* | *list-class* | *invoke-class*
meta-class = **'<'** *class-name* **'>'**
*class-union*⁹ = **'<'** *class-unary* {**'|'** *class-unary*}¹⁺ **'>'**
*invoke-class*¹⁰ = [**'_'** | **'+'**] *parameters*
*list-class*¹¹ = **List** **'{'** ws [*class-desc* ws] **'}'**

¹ Compiler gives warning if *bind* used in *code-block* of a *closure* since it will be binding to captured variable not original variable in surrounding context.

² [Stylistically prefer no ws prior to **'::'** - though not enforcing it via compiler.]

³ Compiler *hint* that expression evaluates to specified class - otherwise error. *class-desc* optional if desired type can be inferred. If *expression* is *variable-identifier* then parser updates type context. [Debug: runtime ensures class specified is received.]

⁴ Explicit conversion to specified class. *class-name* optional if desired type inferable. Ex: **42>>String** calls convert method **Integer@String()** i.e. **42.String()** - whereas **"hello">>String** generates no extra code and is equivalent to **"hello"**.

⁵ Optional *class-desc* is return class - if type not specified **Object** is inferred (or **Boolean** type for predicates or **Auto_** type for closures) for nested parameters / code blocks and **InvokedCoroutine** is inferred for coroutine parameters.

⁶ The optional *binding* indicates the parameter has a default argument (i.e. supplied *expression*) when argument is omitted.

⁷ If optional *class-desc* is omitted **Object** is inferred or **Auto_** for closures or **Boolean** if *variable-name* ends with **'?'**. If *variable-name* ends with **'?'** and *class-desc* is specified it must be **Boolean**.

⁸ **Object** inferred if no classes specified. Class of resulting list bound to *instance-name* is class union of all classes specified.

⁹ Indicates that the class is any one of the classes specified and which in particular is not known at compile time.

¹⁰ **'_'** indicates durational (like coroutine), **'+'** indicates durational/immediate and lack of either indicates immediate (like method). Class **Closure** matches any closure interface. Identifiers and defaults used for parameterless closure arguments.

¹¹ **List** is any **List** derived class. If *class-desc* in item class descriptor is omitted, **Object** is inferred when used as a type or the item type is deduced when used with a *list-literal*. A *list-class* of any item type can be passed to a simple untyped **List** class.

Whitespace:

`wsr`¹ = `{ whitespace }1+`
`ws` = `{ whitespace }`
`whitespace` = `whitespace-char | comment`
`whitespace-char` = `' ' | formfeed | newline | carriage-return | horiz-tab | vert-tab`
`end-of-line` = `newline | carriage-return | end-of-file`
`comment` = `single-comment | multi-comment`
`single-comment` = `'/' { printable } end-of-line`
`multi-comment` = `['*'] { printable } [multi-comment { printable }] ['*']`

Characters and Digits:

`character` = `escape-sequence | printable`
`escape-sequence`² = `'\ ' integer-literal | printable`
`alphanumeric` = `alphabetic | digit | '_'`
`alphabetic` = `uppercase | lowercase`
`lowercase` = `'a' | ... | 'z'`
`uppercase` = `'A' | ... | 'Z'`
`digits` = `'0' | (non-zero-digit {digit})`
`digit` = `'0' | non-zero-digit`
`non-zero-digit` = `'1' | '2' | '3' | '4' | '5' | '6' | '7' | '8' | '9'`
`big-digit` = `digit | alphabetic`

¹ `wsr` is an abbreviation for (w)hite (s)pace (r)quired.

² Special escape characters: `'n'` - newline, `'t'` - tab, `'v'` - vertical tab, `'b'` - backspace, `'r'` - carriage return, `'f'` - formfeed, and `'a'` - alert. All other characters resolve to the same character including `'\'`, `'"`, and `'"`.